Semiconductor Laser Rate Equation Solver

©2011 Krishna Myneni

24 February 2011

1 Introduction

The purpose of this Forth program is to solve the coupled nonlinear rate equations describing the complex electric field and the carrier density in a simple model of the semiconductor laser. The model is sufficient to account for many of the observed dynamics in a *single mode* semiconductor laser in response to a dynamic drive current, such as *relaxation oscillations* and *frequency chirping* [1]. With the addition of terms to represent optical feedback or external optical injection from another laser, the model may be extended to treat more complex dynamics such as *optical feedback induced chaos,* and optical *injection locking* and chaos in coupled lasers. The model treated in this program does not include these external coupling terms, nor does it include more mundane, but important phenomena such as *spontaneous emission noise* and *gain saturation.* Also ignored are phenomena arising explicity from the finite length of the laser (cavity modes) and the waveguide structure. Examples of such phenomena are steady state current tuning of the cavity mode frequency, thermal dissipation and expansion of the laser cavity, and multimode behavior.

The basic quantities to be solved are the complex electric field amplitude, $\tilde{E}(t)$, and carrier density, $N(t)$, inside the semiconductor laser structure, for a given time-dependent injection current, $I(t)$. The field amplitude $\tilde{E}(t)$ is defined by writing the optical electric field of the laser as

$$\varepsilon(t) = E(t)\cos\left(\omega_0 t + \phi(t)\right) = \frac{E(t)}{2}\left(e^{i(\omega_0 t+\phi(t))} + \text{c.c.}\right)$$

where ω_0 is the optical frequency, $E(t)$ is the *real* time-dependent field amplitude, and $\phi(t)$ is the time-dependent phase. The complex field amplitude is,

$$\tilde{E}(t) \equiv E(t) \cdot e^{i\phi(t)}$$

For the simple model of the single mode laser used here, the rate equations may be expressed in terms of the complex field amplitude, without reference to the optical frequency.

The Forth program is based on a similar program written in C [2], but provides greater flexibility in allowing the user to tailor $I(t)$, and the parameters characterizing the laser, from within the Forth environment.

Symbol	Parameter	Representative Value
I_{th}	Threshold current	20 mA
N_{th}	Carrier density at threshold for lasing	$1.5\times10^{18}\mathrm{cm}^{-3}$
τ_p	Photon lifetime	4.5 ps
τ_s	Carrier lifetime	700 ps
α	Linewidth enhancement factor	5
G_N	Differential gain at threshold	$2.6\times10^{-6}\mathrm{cm}^3/\mathrm{s}$

Table 1: Parameters characterizing a semiconductor laser, and typical values of these parameters.

1.1 Requirements

The code should run under any Forth-94 compatible system providing floating point support. Forth systems may use either a separated floating point stack, or an integrated data/fp stack. Modules from the *Forth Scientific Library* [3] are required, and assumed to be in the path, fsl/ , relative to the current path. Required modules include the standard FSL auxiliary files for support of arrays and dynamic memory, fsl-util.x and dynmem.x, the complex arithmetic module, complex.x, and the ordinary differential equation solver, runge4.x.

2a ⟨*ext* 2a⟩≡ (2b)

```
4th
```

2b ⟨*include files* 2b⟩≡ (12)

```
include fsl/fsl-util.⟨ext 2a⟩
include fsl/dynmem.⟨ext 2a⟩
include fsl/complex.⟨ext 2a⟩
include fsl/runge4.⟨ext 2a⟩
```

2 Laser Parameters

The laser structure itself is characterized by the parameters shown in Table 1.

2c ⟨*laser parameters* 2c⟩≡ (12)

```
fvariable I_th     20e     I_th  f!
fvariable N_th     1.5e18  N_th  f!
fvariable t_p      4.5e-12 t_p   f!
fvariable t_s      700e-12 t_s   f!
fvariable alpha    5.0e    alpha f!
fvariable G_N      2.6e-6  G_N   f!
```

Two parameters defined from the characteristic parameters, above, will also prove to be convenient in expressing the dynamical equations for the laser. These are the ratio of the carrier to photon lifetime in the laser,

$$T \equiv \frac{\tau_s}{\tau_p}$$

and the pumping factor,

$$P_0 \equiv \frac{\tau_p G_N N_{\mathrm{th}}}{2}$$

3 ⟨*derived parameters* 3⟩≡ (12)

```
fvariable T_ratio       \ t_s/t_p
fvariable PumpFactor    \ t_p*G_N*N_th/2

: init-params ( - )    \ Initialize the derived parameters
    t_s F@ t_p F@ F/ T_ratio F!
    t_p F@ G_N F@ F* N_th F@ F* 2e F/ PumpFactor F!
;
```

3 Rate Equations

The rate equations describe the time rate of change of the complex electric field amplitude, $\tilde{E}(t)$, and the carrier density, $N(t)$. In the *linear gain approximation*, these equations may be written as,

$$\begin{aligned} \frac{d\tilde{E}}{dt} &= \frac{1}{2}(1+i\alpha)G_N\left(N-N_{\mathrm{th}}\right)\tilde{E} \\ \frac{dN}{dt} &= \frac{I}{eV_a}-\frac{N}{\tau_s}-\left[\frac{1}{\tau_p}+G_N\left(N-N_{\mathrm{th}}\right)\right]\left|\tilde{E}\right|^2 \end{aligned} \tag{1}$$

where e is the electronic charge, and V_a is the active volume of the laser. All other quantities have been defined earlier. A terse derivation of the above equations is given by [4][1]. The above equations are not expressed in SI units. In particular, they have been derived with units of $|\tilde{E}|$ in $\sqrt{n_{\mathrm{ph}}}/\mathrm{cm}^3$, where n_{ph} is the number of photons in the laser cavity. We will not bother to transform eqns. 1 to SI units, since further scaling will be used to transform the dynamical variables to dimensionless quantities.

Exercise: Find the *steady state solutions* of equations 1, i.e., solutions of,

$$\begin{aligned} \frac{1}{2}(1+i\alpha)G_N\left(N^{\mathrm{ss}}-N_{\mathrm{th}}\right)\tilde{E}^{\mathrm{ss}} &= 0 \\ \frac{I^{\mathrm{ss}}}{eV_a}-\frac{N^{\mathrm{ss}}}{\tau_s}-\left[\frac{1}{\tau_p}+G_N\left(N^{\mathrm{ss}}-N_{\mathrm{th}}\right)\right]\left|\tilde{E}^{\mathrm{ss}}\right|^2 &= 0 \end{aligned} \tag{2}$$

For an arbitrary steady-state current, I^{ss}, what must be the value of the steady-state carrier density, N^{ss}? Consider both cases, $I^{\mathrm{ss}} < I_{\mathrm{th}}$ and $I^{\mathrm{ss}} > I_{\mathrm{th}}$, noting that in the former case, $\tilde{E}^{\mathrm{ss}} = 0$, since threshold current is defined to be the current at which lasing starts.

3.1 Normalized Form of the Rate Equations

For numerical solutions of the rate equations, it is convenient to scale the dynamical variables, $\tilde{E}$ and N, and the time, t, to dimensionless form. The dimensionless time, s, is defined by,

$$s \equiv \frac{t}{\tau_p}$$

4 ⟨*s to ns* 4⟩≡ (8d)

```
\ Convert dimensionless time s to nanoseconds
: >ns ( F: s - t )  t_p f@ f* 1e-9 f/ ;
```

[1] We need a reference with accompanying exposition.

The variables $\tilde{E}(t)$ and $N(t)$ are scaled to the dimensionless quantities, $\tilde{Y}(s)$, the normalized complex field amplitude, and $Z(s)$, the normalized carrier density above threshold:

$$\tilde{Y} \equiv \sqrt{\frac{\tau_s G_N}{2}}\tilde{E} \tag{3}$$

$$Z \equiv \frac{\tau_p G_N}{2}\left(N - N_{\text{th}}\right) \tag{4}$$

The coupled rate equations, expressed in normalized form, are[5],

$$\begin{aligned}\frac{d\tilde{Y}}{ds} &= (1+i\alpha)Z(s)\tilde{Y}(s)\\ \frac{dZ}{ds} &= \frac{1}{T}\left(P(s) - Z(s) - (1+2Z(s))\left|\tilde{Y}(s)\right|^2\right)\end{aligned} \tag{5}$$

Exercise: Derive the normalized form of the rate equations, eqns. 5, from the rate equations for $\tilde{E}(t)$ and $N(t)$ (eqns. 1). *Hint:* Use the results from the previous exercise to eliminate the product, eV_a, from the equations.

The two normalized rate equations, expressed in code, are

5a ⟨*dY over ds* 5a⟩≡ (7a)

```
( F: Yre Yim Z - dY/ds )  z*f 1e alpha f@ z*
```

5b ⟨*dZ over ds* 5b⟩≡ (7a)

```
( F: s Z Yre Yim - dZ/ds )
|z|^2 fover 2e f* 1e f+ f* \ F: s Z r
f+ fswap P(s) fswap f-
T_ratio f@ f/
```

3.2 Normalized Pump Rate

In the normalized rate equations 5, the normalized pump rate, $P(s)$, is given by,

$$P = P_0\left(\frac{I(s)}{I_{\text{th}}} - 1\right) \tag{6}$$

where I is the injection current as a function of time, to be defined by the user,

5c ⟨*I(t)* 5c⟩≡ (8d)

```
\ For t in ns, return the injection current
Defer I(t)
```

5d ⟨*P(s)* 5d⟩≡ (7a)

```
\ Compute the pump rate at time s
: P(s) ( F: s - P ) >ns I(t) I_th f@ f/ 1e f- PumpFactor f@ f* ;
```

3.3 The State Vector

The normalized rate equations given in 5 describe the rate of change of a set of three *real* time-dependent physical quantities, $Re\{\tilde{Y}\}$, $Im\{\tilde{Y}\}$, Z. It is convenient to define a *state vector* to track these three real quantities:

6a ⟨*state vector* 6a⟩≡ (12)

```
3 constant SVSIZE
SVSIZE float array sv{
```

It is easier to interpret the complex field amplitude, $\tilde{Y}$, in terms of laser *intensity, W,* and laser *phase*, ϕ, so a transformation of the state vector to intensity and phase will be useful,

$$W \propto \left|\tilde{Y}\right|^2$$

$$\phi = \arg(\tilde{Y})$$

6b ⟨*sv to i phi* 6b⟩≡ (10)

```
\ Compute intensity and phase from the state vector
: intensity ( 'v - ) ( F: - I )   0 } z@ |z|^2 ;
: phase ( 'v - ) ( F: - phase )   0 } z@ arg ; \ phase in radians
```

The transformed components of the state vector, $\{W, \phi, Z\}$ may be computed and printed, using,

6c ⟨*print sv* 6c⟩≡ (10)

```
: print-sv ( 'v - )
    dup intensity   fs. 2 spaces
    dup phase pi f/ fs. 2 spaces    \ normalized phase, 1.0 = pi
    2 } f@          fs. cr          \ normalized carrier density
;
```

Note that the printed phase is $\phi' = \phi/\pi$, i.e. normalized to π radians.

3.4 Frequency Chirp of the Laser

The time derivative of $\phi(t)$ gives the instantaneous frequency change (chirp) of the laser, $\Delta\omega(t)$, with respect to the steady state frequency,

$$\Delta\omega = \omega(N) - \omega(N_{\text{th}}) \tag{7}$$

The program does not calculate the frequency chirp of the laser; however, it may be calculated from the output phase. No attempt is made by the present calculation to unwrap the phase, i.e. at $\phi = \pm\pi$, the phase undergoes a discontinuity since the arg() function is restricted to this range (or to the range, 0-2π, depending on the setting of `PRINCIPAL-ARG` in `complex.x`). A calculation of $\Delta\omega$ requires the output phase be unwrapped prior to computing the derivative.

3.5 Derivative of the State Vector

The set of rate equations are given by the time derivative of the state vector,

$$\left\{ \frac{d\Re\tilde{Y}}{ds}, \frac{d\Im\tilde{Y}}{ds}, \frac{dZ}{ds} \right\}$$

These derivatives are computed and stored in an array, at each step in time, for use by the ODE solver,

7a ⟨*rate equations* 7a⟩≡ (12)

```
⟨P(s) 5d⟩
\ 'u is the state vector array, and 'dudt is the array
\ of computed derivatives
: derivs-sl ( 'u 'du/ds - ) ( F: s - )
      \ or ( s 'u 'du/ds - )
    >r >r
    r@ 0 } z@  r@ 2 } f@ ⟨dY over ds 5a⟩  2r@ drop 0 } z!
    r@ 2 } f@  r@ 0 } z@ ⟨dZ over ds 5b⟩  2r> drop 2 } f!
;
```

4 Injection Current Profile

The time dependent injection current profile, $I(t)$, may be defined by the user. A few cases which are illustrative of different phenomena exhibited by the semiconductor laser may be observed from numerical solutions of the rate equations.

4.1 Constant Current: Below Threshold

The simplest case is to set the injection current to some constant value, I_{dc}, below the laser threshold current, $I_{\mathrm{dc}} < I_{\mathrm{th}}$. The user may verify, either from analysis of the rate equations, or by numerical solution, that for this case, the variables decay from their initial values to the long time limit values, revealed in the exercise from section 3.

7b ⟨*below threshold* 7b⟩≡ (8d)

```
: BelowThreshold ( F: t - I ) fdrop I_th f@ 0.9e f* ;
```

4.2 Constant Current: Above Threshold

The next more interesting case is to choose $I_{\mathrm{dc}} > I_{\mathrm{th}}$, and from the solution of eqns. 2, the values of $\tilde{Y}^{\mathrm{SS}}$and $Z^{\mathrm{SS}}$are known. However, as the laser current is initially switched from $I = 0$ to I_{dc}, both $\tilde{Y}(s)$ and $Z(s)$ display a transient phenomenon known as relaxation oscillations, which eventually damp out to reach the steady state values.

7c ⟨*above threshold* 7c⟩≡ (8d)

```
: AboveThreshold ( F: t - I ) fdrop I_th f@ 1.2e f* ;
```

4.3 Gaussian Current Pulse

Finally we consider the case of a constant current superimposed with a rapidly varying current. Consider a Gaussian current pulse, of width Γ (full width at half maximum), and peak current, I_{p}, superimposed on the d.c. current, I_{dc}. The injection current profile is given by,

$$I(t) = I_{\mathrm{dc}} + I_{\mathrm{p}} \cdot \exp\left(-4\ln(2)\frac{\left(t - t_{\mathrm{offs}}\right)^2}{\Gamma^2}\right)$$

8a ⟨*Gaussian Pulse Parameters* 8a⟩≡ (8d)

```
fvariable fwhm        \ full width at half-max for current pulse
fvariable I_p         \ peak pulse current (above d.c. level)
fvariable I_dc        \ d.c. current level
fvariable t_offs      \ offset time for current peak
```

8b ⟨*Gaussian Pulse* 8b⟩≡ (8d)

```
-4e 2e fln f* fconstant -4ln2

: GaussianPulse ( F: t - I )
    t_offs f@ f- fwhm f@ f/
    fdup f* -4ln2 f* fexp
    I_p f@ f* I_dc f@ f+
;
```

Default settings for the pulse parameters are,

8c ⟨*Gaussian Pulse Default Values* 8c⟩≡ (8d)

```
1e  fwhm f!      \ 1 ns
20e I_p  f!      \ 20 mA
I_th f@ 10e f+ I_dc f!  \ 10 mA above threshold current
3e  t_offs f!    \ pulse peak occurs at 3 ns
```

The default injection current profile will be set to the dc current plus the Gaussian pulse, using the default pulse parameters,

8d ⟨*injection current profile* 8d⟩≡ (12)

```
⟨s to ns 4⟩
⟨I(t) 5c⟩
⟨Gaussian Pulse Parameters 8a⟩
⟨Gaussian Pulse 8b⟩
⟨Gaussian Pulse Default Values 8c⟩
' GaussianPulse is I(t)

\ Alternate profiles

⟨below threshold 7b⟩
⟨above threshold 7c⟩
\ ' BelowThreshold is I(t)
\ ' AboveThreshold is I(t)
```

5 Rate Equation Solver

The rate equation solver for $\tilde{Y}(s)$, $Z(s)$, may now be written. The output of the solver is the time, injection current, and the transformed components of the state vector, computed at discrete time steps. The rate equations, 5, are integrated using the fourth order Runge-Kutta method[6]. While an adaptive step size is much more efficient for integration, a fixed step size is used here in anticipation of extending these calculations to other problems in which the state vector needs to be computed at exact, finely-spaced time intervals[2]. Since the shortest time scale for the dynamical problem is set by τ_p (see table 1), a time step of $\tau_p/10$ is quite sufficient to accurately track the time dependence of $\tilde{Y}$ and Z.

9a ⟨*time step* 9a⟩≡ (10)

```
fvariable ds   \ dimensionless time step
0.1e ds f!     \ actual time step (s) = t_p*ds
```

Integration of the rate equations over a single time step,

$$\tilde{Y}(s), Z(s) \rightarrow \tilde{Y}(s+\Delta s), Z(s+\Delta s)$$

is performed by the following code, which updates the values of $\tilde{Y}$ and Z in the state vector.

9b ⟨*integrate one time step* 9b⟩≡ (10)

```
( F: s - s+ds )
  ds f@ sv{ 1 runge_kutta4_integrate()
```

The state vector is always initialized at the beginning of the solver to its $t = s = 0$ value: $\left|\tilde{Y}(0)\right|^2 = 2$, $\phi(0) = 0$, $N(0) = N_{\text{th}}$.

9c ⟨*initialize state vector* 9c⟩≡ (10)

```
\ initial values of Re{Y}, Im{Y}, Z
2e fsqrt 0e 0e  3 sv{ }fput
```

[2]One such problem is the Lang-Kobayashi model of a semiconductor laser with optical feedback.

Note that if $\tilde{Y}(0) = 0$, $\tilde{Y}(s) = 0$ for all s (see eqns. 5), so a non-zero value of $\tilde{Y}(0)$ is needed to start the dynamics. Consider the source of the non-zero field in a real laser at the onset of lasing.

Once the initial values are set, the rate equation solver will compute and output the laser intensity (arbitrary units), laser phase (units of π), and normalized carrier density (dimensionless) for 20,000 normalized time steps. The total elapsed time is $t_f = 20,000\tau_p\Delta s$. For example, using $\tau_p = 4.5$ ps and $\Delta s = 0.1$, the final time will be $t_f = 9$ ns.

10 ⟨*solver* 10⟩≡ (12)

```
⟨sv to i phi 6b⟩
⟨print sv 6c⟩
⟨time step 9a⟩

: sl ( - )
    init-params                    \ compute all derived parameters
    ⟨initialize state vector 9c⟩
    use( derivs-sl 3 )runge_kutta4_init
    0e                             \ F: - s0
    20000 0 DO                     \ compute 20000 normalized time steps
      fdup >ns fdup f. 2 spaces  \ output real-time in ns
        I(t) f. 2 spaces               \ compute and output injection current
      ⟨integrate one time step 9b⟩
          I 1 mod 0= IF  sv{ print-sv  THEN
   LOOP
   fdrop  runge_kutta4_done
;
```

No.	Physical Quantity
1	Time (ns)
2	Injection Current (mA)
3	Output Intensity (arb)
4	Phase (units of π radians)
5	Normalized Carrier Density Above Threshold

Table 2: Columns Output by `sl`

6 User Interface

All relevant parameters may be displayed by the word, `params.` . Individual laser parameters may be changed by storing new values in the respective variables. Derived parameters will be re-computed automatically at the start of the calculation. The main calculation may be executed by simply typing, `sl` , at the Forth prompt. The calculation will output the state of the laser at the discrete time steps to the console. The output columns are described in table 2. Some Forth systems provide a command to allow output to be redirected to a file. In the absence of a redirection command, the `script` command in Linux may be used to record console output to a log file.

11 ⟨*display parameters* 11⟩≡ (12)

```
: separator ( - ) ." ===================================================" ;
: tab 9 emit ;
: params. ( - )
    cr
    separator cr
    ." Symbol" tab ." Parameter                     " tab ." Value"  cr
    separator cr cr
    ."  t_p  " tab ." Photon lifetime  (s):         " tab  t_p   f@ fs. cr
    ."  t_s  " tab ." Carrier lifetime (s):         " tab  t_s   f@ fs. cr
    ."  G_N  " tab ." Differential gain (cm^3/s):   " tab  G_N   f@ fs. cr
    ."  N_th " tab ." Thr. carrier density (cm^-3): " tab  N_th  f@ fs. cr
    ."  I_th " tab ." Thr. current (mA):            " tab  I_th  f@ f. cr
    ."  alpha" tab ." Linewidth enhancement factor: " tab  alpha f@ f. cr
    separator cr
    ." Derived Dimensionless Parameters " cr
    separator cr cr
    ." t_s/t_p ratio: " tab T_ratio    f@ f. cr
    ." Pump factor: "   tab PumpFactor f@ f. cr
    separator cr
;
```

7 Main Program

The main program is assembled.

12 ⟨*sll.fs* 12⟩≡
  ⟨*include files* 2b⟩
  ⟨*laser parameters* 2c⟩
  ⟨*derived parameters* 3⟩
  ⟨*display parameters* 11⟩
  ⟨*injection current profile* 8d⟩
  ⟨*state vector* 6a⟩
  ⟨*rate equations* 7a⟩
  ⟨*solver* 10⟩

```
init-params
params.
```

8 Version History

- **2011-02-24** *KM*, fixed typos and punctuation; reordered some text.
- **2011-02-22** *KM* rewritten in the literate programming style using LyX; corrected a mistake in the value of $4\ln(2)$, which was hard-coded as `2.77066e;` added constant current injection profiles.
- **2007-10-19** *KM* modified to use complex library, and FSL ODE solver. This version is about 20% slower than the original, but the code simplifies greatly.
- **2002-10-27** *KM* changed all instances of dfloat to float for ANS Forth portability. Removed explicit fp number size dependence.
- **2002-10-24** *KM* fixed problem with the main loop; previously was not computing Vdot on every loop iteration. Also changed current pulse pos to 3 ns.
- **2002-10-21** *KM* fixed time scale problem in `sl` after problem was pointed out by Marcel Hendrix.
- **2000-01-26** *KM* first version, based on [2].

References

[1] G. P. Agrawal and N. K. Dutta, *Semiconductor Lasers*, New York: Van Nostrand Reinhold (1993).

[2] S. D. Pethel, *C program for numerical solution of the semiconductor laser rate equations*, unpublished (2000).

[3] Forth Scientific Library home page at `http://www.taygeta.com/fsl/sciforth.html` ; compatible modules are also provided at `ftp://ccreweb.org/software/fsl/`

[4] K. Myneni, *Phenomenological rate equations for a semiconductor laser*, `http://ccreweb.org/documents/physics/amo/sl/rateqns.html` (2008).

[5] D. W. Sukow, *Experimental Control of Instabilities and Chaos in Fast Dynamical Systems*, PhD Thesis, Duke University (1997).

[6] W. H. Press, S. A. Teukolsky, W. T. Vetterling, and B. P. Flannery, *Numerical Recipes in C, 2nd ed.*, Cambridge University Press (1994).